Comparative Statics

LaTeX file: *ComparativeStatics-nb-all* — Daniel A. Graham <daniel.graham@duke.edu>, June 22, 2005

Comparative-statics involves the determination of the effect of changes in the value of an exogenous variables on the value of an endogenous variable , e.g., the effect of a change in price of x on the quantity demanded of x.

Explicit Solution

If we have enough information we can solve for the comparative static effects explicitly.

If, for example, we started with the Cobb-Douglas utility function

In: $\text{util} = a\ \text{Log}[x] + b\text{Log}[y];$

we could derive the demand function for x via the substitution method by solving the budget constraint for y,

In: $\text{budget} = \text{Solve}[\text{px}\ x + \text{py}\ y == m, y]$

Out: $\left\{\left\{y \to \frac{m - \text{px}\ x}{\text{py}}\right\}\right\}$

substituting the result into the utility function,

In: $\text{util} = \text{util}/.\text{budget}[[1]]$

Out: $a\text{Log}[x] + b\text{Log}\left[\frac{m - \text{px}\ x}{\text{py}}\right]$

differentiating with respect to x,

In: $\text{firstorder} = D[\text{util}, x]$

Out: $\frac{a}{x} - \frac{b\text{px}}{m - \text{px}\ x}$

setting the result equal to zero and solving for x

In: $\text{sol} = \text{Solve}[\text{firstorder} == 0, x]$

Out: $\left\{\left\{x \to \frac{am}{(a+b)\ \text{px}}\right\}\right\}$

We could then differentiate the demand function obtained in this way with respect to px to obtain the comparative-static effect of a change in px upon the quantity demanded of x:

In: $D[x/.\text{sol}[[1]], \text{px}]$

Out: $-\frac{am}{(a+b)\ \text{px}^2}$

In: $\text{InputForm}[\text{Simplify}[\%]]$

Out: $-((a * m)/((a+b) * \text{px}\hat{\ }2))$

Note that starting with an explicit utility function, we obtain an equally explicit expression for the comparative-static effect. Starting with a slightly less explicit utility function, say u = a Log[x] + b Log[y] we would have arrived at a slightly less explicit comparative-static effect which includes the (unknown) parameters a and b.   There is a general principle here that I call conservation-of-information: the more information you begin with, the more you'll end with.

Comparative-static analysis in Economics usually begins with still less explicit information and the best that one can hope for in such circumstances is to determine the sign, positive or negative, of the comparative-static effect.   There are two, time-honored methods for doing this which I call Brute-Force and Cheap-Trick.

Brute-Force

Carry out the following steps:

In: util = $u[x] + y$/.Solve [px x + py y == m, y] [[1]]

Out: $\frac{m - \text{px}\, x}{\text{py}} + u[x]$

In: firstorder = D [util, x] == 0/.$x \to x$ [px, py, m]

Out: $-\frac{\text{px}}{\text{py}} + u'[x[\text{px}, \text{py}, m]] == 0$

In: secondorder = D [util, x, x] <=0/.$x \to x$ [px, py, m]

Out: $u''[x[\text{px}, \text{py}, m]] \leq 0$

The substitution of x[px,py,m] for x is done to acknowledge explicitly the dependence of the endogenous variable x upon the exogenous variables px, py and m.   "firstorder" is a necessary condition for an interior maximum, x > 0 and y > 0, and "secondorder" is a sufficient condition for a maximum rather than a minimum.   Since "firstorder" is an identity in the exogenous variables provided that they give rise to an interior solution, it can be differentiated with respect to any of them:

In: compstat = D [firstorder, px]

Out: $-\frac{1}{\text{py}} + u''[x[\text{px}, \text{py}, m]]\, x^{(1,0,0)}[\text{px}, \text{py}, m] == 0$

In: compstathw = D [firstorder, py]

Out: $\frac{\text{px}}{\text{py}^2} + u''[x[\text{px}, \text{py}, m]]\, x^{(0,1,0)}[\text{px}, \text{py}, m] == 0$

Note that $x^{(0,1,0)}$ is *Mathematica*'s notation for the partial derivative of x with respect to the second argument. The result can be solved for the comparative static effect

In: sol = Solve [compstat, D [x [px, py, m] , px]]

Out: $\left\{\left\{x^{(1,0,0)}[\text{px}, \text{py}, m] \to \frac{1}{\text{py}\, u''[x[\text{px}, \text{py}, m]]}\right\}\right\}$

In: solhw = Solve [compstathw, D [x [px, py, m] , py]]

Out: $\left\{\left\{x^{(0,1,0)}[\text{px}, \text{py}, m] \to -\frac{\text{px}}{\text{py}^2 u''[x[\text{px}, \text{py}, m]]}\right\}\right\}$

In: InputForm [Simplify [D [x [px, py, m] , py] /.solhw [[1]]]]

Out: − (px/ (py^2 * Derivative [2] [u] [x [px, py, m]]))

Based on the above results, including "secondorder", we can conclude that the sign of D[x[px,py,m],px] at an interior solution is negative.

Cheap-Trick

To understand this approach get some paper and draw an x-axis and a y-axis. Now think about drawing a graph of D[util,x] for "util" defined above where you regard everything but x as a constant. What will the graph look like? Well, if it crosses the x-axis to the right of the origin, "firstorder" is satisfied and if it crosses from above to below in moving from left to right, then "secondorder" is satisfied as well. Without worrying about being accurate, draw such a graph. The crossing of the x-axis is an interior solution for an optimal x. The key to "cheap-trick" is to ask what effect a change in px would have upon the graph you just drew. If it shifts it up, then the point where it crosses the x-axis and thus the optimal value of x would shift to the right. If it shifts it down, then the point where it crosses the x-axis and thus the optimal value for x would shift to the left. All that remains is to note that determining how the graph changes with px is just a matter of differentiating D[util,x] with respect to px - nothing could be simpler! Here's the proposition:

Sign[D[x[px,py,m],px]] = Sign[D[util,x,px]]

Here's the application to the current problem. Since

In: Sign [D [util, x, px]]

Out: $-\frac{1}{\text{Sign}[\text{py}]}$

In: InputForm [Simplify [Sign [D [util, x, px]]]]

Out: −Sign [py] ^ (−1)

"cheap-trick" also implies that the sign of D[x[px,py,m],px] at an interior solution is negative — but notice how much "cheaper" it is!